******************************************
*
*	Article: The Human Costs of the War on Drugs: Attitudes towards Militarization of Security in Mexico
*	Authors: Juan Masullo J., Davide Morisi
*	Journal: Comparative Political Studies
*	Date: 23 March 2023
*
*	This file replicates the following
*	RECODING STUDY 3
*
*	All analyses performed with Stata/SE 14.2 for Windows
******************************************

*Set working directory
*[my working directory]

*Open dataset
use "Replication_CPS\Data\study3.dta", clear


/*DV
Q18 En una escala del 0 al 10, donde 0 significa “muy desfavorable” y 10 “muy favorable”,
¿qué tan favorable o desfavorable es su opinión sobre el uso de las fuerzas militares
en el combate contra el crimen organizado?
o	0 (muy desfavorable)  (1) 
o	1  (2) 
o	2  (3) 
o	3  (4) 
o	4  (5) 
o	5 (ni desfavorable ni favorable)  (6) 
o	6  (7) 
o	7  (8) 
o	8  (9) 
o	9  (10) 
o	10 (muy favorable)  (11) */
fre Q18
gen dv_att = (Q18-1)/10 // rescaled from 0 to 1
label var dv_att "How in favor of using army to combat organized crime"
label de agree 0"strongly against" 1"strongly in favor"
label val dv_att agree
fre dv_att
*standardized
egen st_dv_att = std(dv_att)


*******
*trust
*En una escala del 0 al 10, donde 0 significa "nada confianza" y 10 "mucha confianza", ¿cuánta confianza tiene usted en...

*trust in army
label de trust 0"no trust" 10"high trust"
label val trust_army trust
fre trust_army
gen trust_army01 = trust_army/10 // rescaled from 0 to 1


*attention check
fre attention_check
recode attention_check (2 3=1 "correct") (else=0 "not correct"), gen(attention_correct)
fre attention_correct


*************
*victime of crime

*general
fre victim_crime
*Q51 Ahora, cambiando el tema, ¿En los últimos 12 meses ha sido usted  víctima de algún tipo de crimen? Por ejemplo, robo, agresión, fraude, chantaje,  extorsión, amenazas, o algún otro tipo de acto delincuencial.
*o	Sí  (1) 
*o	No  (2) 
*o	Prefiero no responder  (3) 

*specific
*Q36 ¿Ha sido usted víctima de al menos uno de los siguientes actos criminales o delincuenciales en los últimos 12 meses?

*not related to drug cartel
*Robo (en trasporte público, de autopartes u otro tipo de robo), Agresión, Chantaje o Fraude (1) 
fre victim_specific_not_cartel // 1=yes, 2=no, 3=prefer not to reply
recode victim_specific_not_cartel (2 3=0 "no/no reply") (1=1 "yes, victim") (.=0), gen(victim_not_cartel)
label var victim_not_cartel "Victim of crime not related to drug cartel?"
fre victim_not_cartel

*related to cartel
*Extorsión, Cobro de piso, Secuestro, Asalto con arma de alto calibre, Amenaza de muerte, Narcobloqueo (2) 
fre victim_specific_cartel
recode victim_specific_cartel (2 3=0 "no/no reply") (1=1 "yes, victim") (.=0), gen(victim_cartel)
label var victim_cartel "Victim of crime related to drug cartel?"
fre victim_cartel // 14%


****
*socio demographics

*gender
fre gender
recode gender 1=0 3=0 2=1, gen(female)
label de female 0"male/other" 1"female"
label val female female
fre female

*age
gen age10=age/10


*education
fre education
*recode edu
recode education (1/3=1 "secundaria or below") ///
(4=2 "Preparatoria, bachillerato o equivalente") ///
(5=3 "Técnico superior") ///
(6/7=4 "university") (8=.), gen(edu4)
fre edu4
*recode other category
fre education_other
encode education_other, gen(education_otheren)
replace edu4 = 3 if education_otheren==1


*state - regions
fre state_name
encode state_name, gen(staten)
fre staten
recode staten (2 3 6 10 25 26=1 "North West") ///
(8 19 28=2 "North East") ///
(9 14 16 18=3 "Centre West") ///
(1 11 24 32=4 "Centre") ///
(13 17 21 22 29=5 "Centre South") ///
(7 15=6 "Estado de Mexico,Federal district") ///
(4 5 12 20 23 27 30 31=7 "South"), gen(regions) 
fre regions



************
save "Replication_CPS\Data\study3_recoded.dta", replace

